Computergestuetzte Mathematik zur Analysis

Lektion 10 (7. Januar)

Raumkurven (Wiederholung)

```
> restart: with(plots):
> kurve := (2 - cos(t/6))*cos(t), (2 - cos(t/6))*sin(t), t/8;
```

$$kurve := \left(2-\cos\left(\frac{1}{6}\,t\right)\right)\cos(t),\ \left(2-\cos\left(\frac{1}{6}\,t\right)\right)\sin(t),\ \frac{1}{8}\,t \qquad (1.1)$$

Achtung: kurve ist eine Folge

```
> spacecurve([ kurve, t = -6*Pi .. 6*Pi], numpoints = 300,
  thickness = 5);
```




```
> kurve := [t, k, exp(-k^2*t^2), t = -2 .. 2];
```

(1.2)

$$kurve := [t, k, e^{-k^2 t^2}, t = -2..2] \quad (1.2)$$

```
> kurvenmenge := { seq(kurve, k = 1 .. 7) };
```

$$kurvenmenge := \{[t, 1, e^{-t^2}, t = -2..2], [t, 2, e^{-4t^2}, t = -2..2], [t, 3, e^{-9t^2}, t = -2..2], [t, 4, e^{-16t^2}, t = -2..2], [t, 5, e^{-25t^2}, t = -2..2], [t, 6, e^{-36t^2}, t = -2..2], [t, 7, e^{-49t^2}, t = -2..2]\} \quad (1.3)$$

```
> spacecurve(kurvenmenge, axes =  frame, thickness = 3);
```




```
> kurve := (5+cos(21*t))*cos(2*t), (5+cos(21*t))*sin(2*t), sin
  (21*t);
```

$$kurve := (5 + \cos(21\,t))\cos(2\,t), (5 + \cos(21\,t))\sin(2\,t), \sin(21\,t) \quad (1.4)$$

```
> spacecurve([kurve, t = 0 .. 2*Pi],numpoints=500,thickness=3);
```




```
> n := 500; j:='j'
```

$$n := 500$$

Warning, inserted missing semicolon at end of statement

$$j := j \tag{1.5}$$

```
> rgb_wert := evalf(sin(j*Pi/n)^2), 0, evalf(cos(j*Pi/n)^2);
```

$$rgb\_wert := \sin(0.006283185308\,j)^2,\ 0,\ \cos(0.006283185308\,j)^2 \tag{1.6}$$

```
> for j from 1 to n do;
>   p1 := subs(t = (j-1)*2*Pi/n, [kurve]);
>   p2 := subs(t = j*2*Pi/n, [kurve]);
>   pl[j] := spacecurve( [p1, p2], color = COLOR(RGB, rgb_wert),
  thickness = 3):
> od:
> display(convert(pl, set));
```



Flaechen im Raum (Wiederholung)

```
> restart:
> profil :=  cosh(t); #
```

$$profil := \cosh(t) \tag{2.1}$$

```
> plot(cosh(t),t=-1..1);
```




```
> flaeche := [ t, cos(s)*profil, sin(s)*profil];
```

$$flaeche := [t, \cos(s)\cosh(t), \sin(s)\cosh(t)] \quad \textbf{(2.2)}$$

```
> plot3d(flaeche, s = 0 .. 2*Pi, t = -1 .. 1);
```




```
> plot3d(flaeche, s = 0 .. 2*Pi, t = -1 .. 1,color="DarkGreen",
  style=patchnogrid,lightmodel=light4,glossiness=0.1,viewpoint=
  "circleleft",orientation=[30,45]);
```



Partielle Ableitungen

```
> f := exp(x); with(plots):
```

$$f := e^{x} \tag{3.1}$$

```
> df := Diff(f, x);
```

$$df := \frac{d}{dx} e^{x} \tag{3.2}$$

```
> value(df);
```

$$e^{x} \tag{3.3}$$

```
> g := exp(a*x + b*y + c*z);
```

$$g := e^{a x + b y + c z} \tag{3.4}$$

```
> dg := Diff(g, x);
```

$$dg := \frac{\partial}{\partial x} e^{a x + b y + c z} \tag{3.5}$$

```
> value(dg);
```

(3.6)

$$a\,e^{a x + b y + c z} \tag{3.6}$$

```
> d123g := Diff(g, x, y, y, z$3);
```

$$d123g := \frac{\partial^6}{\partial z^3\,\partial y^2\,\partial x}\,e^{a x + b y + c z} \tag{3.7}$$

```
> value(d123g);
```

$$a\,b^2\,c^3\,e^{a x + b y + c z} \tag{3.8}$$

```
> h := (x, y, z) -> sin(a*x + b*y + c*z);
```

$$h := (x, y, z) \to \sin(a\,x + b\,y + c\,z) \tag{3.9}$$

```
> D[2](h);
```

$$(x, y, z) \to \cos(a\,x + b\,y + c\,z)\,b \tag{3.10}$$

```
> D[1, 2, 2, 3$3](h);
```

$$(x, y, z) \to -\sin(a\,x + b\,y + c\,z)\,a\,b^2\,c^3 \tag{3.11}$$

```
> f := (x, y) -> sin(sqrt(x^2 + y^2)) * ((x-1/4)^2-(y-1/3)^2);
```

$$f := (x, y) \to \sin\left(\sqrt{x^2 + y^2}\right)\left(\left(x - \frac{1}{4}\right)^2 - \left(y - \frac{1}{3}\right)^2\right) \tag{3.12}$$

```
> p1 := plot3d(f-.05, -2 .. 2, -2 .. 2, style = surfacecontour,
  contours=30, shading = zhue):
> display(p1,orientation=[-40,50]);
```




```
> y_schnittkurve := [t, y, f(t, y), y = -2..2];
```

$$y\_schnittkurve := \left[t, y, \sin\left(\sqrt{t^2+y^2}\right)\left(\left(t-\frac{1}{4}\right)^2-\left(y-\frac{1}{3}\right)^2\right), y=-2..2\right] \quad \textbf{(3.13)}$$

```
> tangente := f(1/2,-1) + D[2](f)(1/2,-1) + D[2](f)(1/2,-1)*y:
> plot([[y,f(1/2,y),y=-2..2],[y,tangente,y=-2..0]],color=[black,
  red], thickness = 3);
```




```
> y_schnitte :=spacecurve({seq(y_schnittkurve, t=-2..2,1/2)},
  color = black, thickness = 3):
> display([p1,y_schnitte],orientation=[-40,50]);
```




```
> p := <-3/2, -1, f(-3/2, -1)>;
```

$$p := \begin{bmatrix} -\frac{3}{2} \\ -1 \\ \frac{185}{144}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix} \tag{3.14}$$

```
> Dy := D[2](f)(-3/2, -1);
```

$$Dy := -\frac{185}{936}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + \frac{8}{3}\sin\left(\frac{1}{2}\sqrt{13}\right) \tag{3.15}$$

```
> y_tan := p + t.<0,1,Dy>;
```

$$y\_tan := t.\begin{bmatrix} 0 \\ 1 \\ -\frac{185}{936}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + \frac{8}{3}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix} \tag{3.16}$$

$$+\begin{bmatrix} -\frac{3}{2} \\ -1 \\ \frac{185}{144}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix}$$

```
> y_tan := simplify(y_tan);
```

$$y\_tan := \begin{bmatrix} -\frac{3}{2} \\ -1+t \\ -\frac{185}{936}t\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}+\frac{8}{3}t\sin\left(\frac{1}{2}\sqrt{13}\right)+\frac{185}{144}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix} \quad (3.17)$$

```
> y_tan_pl := spacecurve(convert(y_tan, list), t = -1 .. 3/2,
  color = red, thickness = 5):
> display({p1,y_schnitte,y_tan_pl}, orientation=[-40,50]);
```



```
> grad := <D[1](f)(-3/2,-1),D[2](f)(-3/2,-1)>;
  ngrad := norm(grad,2):
  dgrad:= simplify(grad/ngrad):
```

$$grad := \begin{bmatrix} -\frac{185}{624}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} - \frac{7}{2}\sin\left(\frac{1}{2}\sqrt{13}\right) \\ -\frac{185}{936}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + \frac{8}{3}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix} \tag{3.18}$$

```
> grad_tan := p+t.<dgrad[1],dgrad[2],ngrad>;
```

$$grad\_tan := t.\left[\left[\left(-555\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} - 6552\sin\left(\frac{1}{2}\sqrt{13}\right)\right)\Big/ \left(5784025 + 62064743\sin\left(\frac{1}{2}\sqrt{13}\right)^2 + 3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\sin\left(\frac{1}{2}\sqrt{13}\right)\right)^{1/2}\right], \left[\left(-370\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + 4992\sin\left(\frac{1}{2}\sqrt{13}\right)\right)\Big/ \left(5784025 + 62064743\sin\left(\frac{1}{2}\sqrt{13}\right)^2 + 3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\sin\left(\frac{1}{2}\sqrt{13}\right)\right)^{1/2}\right], \left[\left(\left(\frac{185}{624}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + \frac{7}{2}\sin\left(\frac{1}{2}\sqrt{13}\right)\right)^2 + \left(-\frac{185}{936}\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + \frac{8}{3}\sin\left(\frac{1}{2}\sqrt{13}\right)\right)^2\right)^{1/2}\right]\right] + \begin{bmatrix} -\frac{3}{2} \\ -1 \\ \frac{185}{144}\sin\left(\frac{1}{2}\sqrt{13}\right) \end{bmatrix} \tag{3.19}$$

```
> grad_tan := simplify(grad_tan);
```

$$grad\_tan := \left[\left[-\frac{3}{2}\left(370\,t\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13} + 4368\,t\sin\left(\frac{1}{2}\sqrt{13}\right) + \left(-62064743\cos\left(\frac{1}{2}\sqrt{13}\right)\right)^2\right.\right. \tag{3.20}$$

$$+3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\,\sin\left(\frac{1}{2}\sqrt{13}\right)+67848768\Bigg)^{1/2}\Bigg)\Bigg/$$

$$\Bigg(-62064743\cos\left(\frac{1}{2}\sqrt{13}\right)^2$$

$$+3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\,\sin\left(\frac{1}{2}\sqrt{13}\right)+67848768\Bigg)^{1/2}\Bigg],$$

$$\Bigg[-\Bigg(370\,t\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}-4992\,t\sin\left(\frac{1}{2}\sqrt{13}\right)$$

$$+\Bigg(-62064743\cos\left(\frac{1}{2}\sqrt{13}\right)^2$$

$$+3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\,\sin\left(\frac{1}{2}\sqrt{13}\right)+67848768\Bigg)^{1/2}\Bigg)\Bigg/$$

$$\Bigg(-62064743\cos\left(\frac{1}{2}\sqrt{13}\right)^2$$

$$+3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\,\sin\left(\frac{1}{2}\sqrt{13}\right)+67848768\Bigg)^{1/2}\Bigg],$$

$$\Bigg[\frac{185}{144}\sin\left(\frac{1}{2}\sqrt{13}\right)$$

$$+\frac{1}{1872}\Bigg(-62064743\cos\left(\frac{1}{2}\sqrt{13}\right)^2$$

$$+3578640\cos\left(\frac{1}{2}\sqrt{13}\right)\sqrt{13}\,\sin\left(\frac{1}{2}\sqrt{13}\right)+67848768\Bigg)^{1/2}t\Bigg]\Bigg]$$

```
> grad_tan_pl :=spacecurve(convert(grad_tan, list), t = 0 .. 3/2,
  color = blue, thickness = 3):
> display({p1,y_schnitte,grad_tan_pl}, orientation=[90,00]);
```



Ableitungen von Vektorfunktionen

```
> restart: with(VectorCalculus):
```

```
> v := <t, t^2, t^3>;
```

$$v := (t)\,e_x + (t^2)\,e_y + (t^3)\,e_z \tag{4.1}$$

```
> diff(v, t):
> with(VectorCalculus):
> diff(v, t);
```

$$e_x + 2\,t e_y + 3\,t^2 e_z \tag{4.2}$$

```
> BasisFormat(false);
```

$$true \tag{4.3}$$

```
> dv := diff(v, t);
```

(4.4)

$$dv := \begin{bmatrix} 1 \\ 2\,t \\ 3\,t^2 \end{bmatrix} \qquad (4.4)$$

```
> with(plots):
> spacecurve(v, t = -3 .. 3,thickness=3);
```



Moebiusband

```
> restart: with(plots):
> M := <cos(t)*(1 + s*cos(t/2)),
        sin(t)*(1+s*cos(t/2)),
        s*sin(t/2)>;
  p1:= plot3d(M, t = 0 .. Pi, s=-1/2..0,color=blue):
  p2:= plot3d(M, t = 0 .. Pi, s=0..1/2,color=red):
  p3:= spacecurve(subs(s=1/2+0.02,convert(M,list)),t=0..Pi,color=
  coral,thickness=5):
  display({p1,p2,p3});
```

$$M := \begin{bmatrix} \cos(t)\left(1 + s\cos\left(\frac{1}{2}t\right)\right) \\ \sin(t)\left(1 + s\cos\left(\frac{1}{2}t\right)\right) \\ s\sin\left(\frac{1}{2}t\right) \end{bmatrix}$$




```
> Seele := subs(s = 0, M);
```

$$Seele := \begin{bmatrix} \cos(t) \\ \sin(t) \\ 0 \end{bmatrix} \quad (5.1)$$

```
> with(VectorCalculus):
> BasisFormat(false);
```

$$true \quad (5.2)$$

```
> Mt := diff(Seele, t);
```

$$Mt := \begin{bmatrix} -\sin(t) \\ \cos(t) \\ 0 \end{bmatrix} \tag{5.3}$$

```
> Ms := diff(M, s);
```

$$Ms := \begin{bmatrix} \cos(t)\cos\left(\frac{1}{2}t\right) \\ \sin(t)\cos\left(\frac{1}{2}t\right) \\ \sin\left(\frac{1}{2}t\right) \end{bmatrix} \tag{5.4}$$

```
> with(LinearAlgebra):
> Normale := CrossProduct(Ms, Mt);
```

$$Normale := \begin{bmatrix} -\sin\left(\frac{1}{2}t\right)\cos(t) \\ -\sin\left(\frac{1}{2}t\right)\sin(t) \\ \cos(t)^2\cos\left(\frac{1}{2}t\right) + \sin(t)^2\cos\left(\frac{1}{2}t\right) \end{bmatrix} \tag{5.5}$$

```
> pl1 := plot3d(M, t = 0 .. 2*Pi, s = -1/3 .. 1/3, grid = [60,
  5], color = red):
> EinheitsNormale := simplify(Normale/Norm(Normale, 2)) assuming
  t::real:
> EinheitsNormale[1];
```

$$-\sin\left(\frac{1}{2}t\right)\left(2\cos\left(\frac{1}{2}t\right)^2 - 1\right) \tag{5.6}$$

```
> flaeche := convert(Seele + s*EinheitsNormale, list);
```

$$flaeche := \left[\cos(t) - s\sin\left(\frac{1}{2}t\right)\left(2\cos\left(\frac{1}{2}t\right)^2 - 1\right), \sin(t) - s\sin\left(\frac{1}{2}t\right)\sin(t), s\cos\left(\frac{1}{2}t\right)\right] \tag{5.7}$$

```
> pl2 := plot3d(flaeche, t = 0 .. 2*Pi, s = 0 .. .4, color = s,
  numpoints = 3000, style = patchnogrid):
> with(plots):
> display({pl1, pl2}, orientation = [-78, -159]);
```


-0.4
-0.2
0
0.2
0.4
-1
-0.5
0
0.5
1
-1
-0.5
0
0.5
1